



Munich Personal RePEc Archive

Introduction to Computational Economywide Modeling with GAMS

Mercado, Ruben and Kendrick, David

The University of Texas at Austin

October 2002

Online at <https://mpra.ub.uni-muenchen.de/111973/>
MPRA Paper No. 111973, posted 16 Feb 2022 16:20 UTC

Introduction to Computational Economywide Modeling with GAMS

by

P. Ruben Mercado

and

David A. Kendrick

(October 2002)

The University of Texas at Austin

The analysis of economywide models is a particularly demanding topic in economics since it involves the study of interdependence. It implies a move to the realm of multiple heterogeneous agents, sectors and institutions interacting in complex ways. While there are some analytical methods and results available to help us in such endeavor, computational methods become necessary when we move to medium to large size models or when we have to deal with particularly complex ones.

Our goal here is to provide a basic introduction to the art of economy-wide modeling. We present a sequence of small models, we show how to implement them in the General Algebraic Modeling System (GAMS)¹ and we perform and suggests some experiments. We start with an Input-Output model in which quantities produced are determined given technology and demand levels, we follow with a Production Prices model which determines relative prices given technology and a distributive variable, and then we move to a General Equilibrium model in which prices and quantities are determined simultaneously given technology, preferences and endowments. Finally, we introduce SAM based and Johansen style Computable General Equilibrium models.²

1. Input-Output Model

A good starting point for the study of interdependence in economics is the well known Input-Output model pioneered by Nobel prize winner Wassily Leontief.³ One of the main goals of this type of model is the determination of direct and indirect levels of production to satisfy a given increase in final demand.

¹ For an introduction to GAMS programming, see Brooke et al. (1998) and Zenios (1996).

² We will present models in a sequence reflecting mainly their computational complexity in terms of degree of non-linearity and size. The order of the sequence does not mean historical or theoretical precedence of one type of model over the others, or a ranking of practical relevance.

³ Leontief (1953).

Imagine that there is an economy with three industries (1, 2 and 3). Each of them produces a single output, using as inputs part or its own production as well as part of the output from the other industries. It is clear, then, that each industry plays a dual role since it is a supplier of inputs and also user of outputs. Imagine that each product in this economy is also used to satisfy an exogenously given level of demand from the part of consumers. In formal terms, we can represent the economy just described as follows:

$$\begin{aligned}x_1 &= a_{11} x_1 + a_{12} x_2 + a_{13} x_3 + d_1 \\x_2 &= a_{21} x_1 + a_{22} x_2 + a_{23} x_3 + d_2 \\x_3 &= a_{31} x_1 + a_{32} x_2 + a_{33} x_3 + d_3\end{aligned}$$

where the x 's are production levels, a_{ij} are the input-output coefficients (the intermediate requirements from industry i per unit of output of industry j), and where the d 's are the levels of final demand from the consumers. In matrix terms, we can write:

$$x = Ax + d$$

where x is the vector of levels of production, d is the vector of final demands and A is the input-output coefficients matrix. An interesting question can be posed to this economy. For example, given an input-output coefficients matrix:

$$A = \begin{bmatrix} 0.3 & 0.2 & 0.2 \\ 0.1 & 0.4 & 0.5 \\ 0.4 & 0.1 & 0.2 \end{bmatrix}$$

and given a vector of final demands:

$$d = \begin{bmatrix} 4 \\ 5 \\ 3 \end{bmatrix}$$

What will be the required level of total production of each industry (direct and indirect) to satisfy that final demand vector? The GAMS representation of this problem is:

```
$TITLE IO-1
* Input-Output Model

SCALARS
d1 final demand for x1 /4/
d2 final demand for x2 /5/
d3 final demand for x3 /3/;

VARIABLES
x1 production level industry 1
```

```

x2 production level industry 2
x3 production level industry 3
j performance index;

EQUATIONS
eqx1
eqx2
eqx3
jd performance index definition;

jd.. j =E= 0;
eqx1.. x1 =E= 0.3*x1 + 0.2*x2 + 0.2*x3 + d1;
eqx2.. x2 =E= 0.1*x1 + 0.4*x2 + 0.5*x3 + d2;
eqx3.. x3 =E= 0.4*x1 + 0.1*x2 + 0.2*x3 + d3;

MODEL IO /jd, eqx1, eqx2, eqx3/;
SOLVE IO MAXIMIZING J USING LP;
DISPLAY x1.l, x2.l, x3.l;

```

and the solution obtained is:

$$x_1 = 16.821, \quad x_2 = 23.744, \quad x_3 = 15.128.$$

There are analytical methods available to deal with this problem.⁴ Indeed, the analytical solution is given by:

$$x = (I - A)^{-1} d$$

where I is the identity matrix. This formula can be easily handled for small models. However, computational methods will be required to perform the matrix inversion as soon as we move to larger models. And these methods will become unavoidable as we move to more complex problems. For example, imagine now that we have some restriction, like a capacity constraint, on the maximum level of production of some products (say $x_2 \leq 22$ and $x_3 \leq 14$) and we want to know the maximum level of demand of product 1 (d_1) that the economy can satisfy, given the demand levels d_2 and d_3 . This can be easily handled in GAMS. Here is the corresponding GAMS representation of the problem:

```

$title IO-2
* Input-Output Model with restrictions

SCALARS
d2 final demand for x2 /5/
d3 final demand for x3 /3/;

POSITIVE VARIABLES
x1 production level industry 1
x2 production level industry 2

```

⁴ See for example Chiang (1984) for an introduction to these methods.

```

x3 production level industry 3
d1 final demand for x1;

VARIABLES
j performance index;

EQUATIONS
eqx1
eqx2
eqx3
res1 restriction 1
res2 restriction 2
jd performance index definition;

jd.. j =E= d1;
eqx1.. x1 =E= 0.3*x1 + 0.2*x2 + 0.2*x3 + d1;
eqx2.. x2 =E= 0.1*x1 + 0.4*x2 + 0.5*x3 + d2;
eqx3.. x3 =E= 0.4*x1 + 0.1*x2 + 0.2*x3 + d3;
res1.. x2 =L= 22;
res2.. x3 =L= 14;

MODEL IO /all/;
SOLVE IO MAXIMIZING j USING LP;
DISPLAY x1.l, x2.l, x3.l, d1.l;

```

Notice that we just had to define and add two equations (*res1* and *res2*) corresponding to the restrictions, set the performance index equal to d_1 , and define d_1 as a variable (no longer as a scalar). Also, to avoid negative values which make no economic sense we had to define all variables but the performance index as positive variables. Solving the problem, we obtain:

$$x_1 = 14.143, \quad x_2 = 22, \quad x_3 = 13.571, \quad d_1 = 2.786$$

which indicates that the maximum level of demand of good 1 that can be achieved given the restrictions is equal to 2.786, obviously lower than in the previous example since we set the values of the restrictions below the solution levels previously obtained. On the contrary, if the economy is able to lift those “bottlenecks” up to 30 for x_2 and 20 for x_3 , the demand of goods produced by sector 1 that could be satisfied will be $d_1 = 7.8$.

2. Production Prices Model

So far we have been dealing with a model with two main type of agents (consumers and industries), in which their interrelations are linear and where, given a technology (the input-output coefficients matrix) we determine quantities produced and/or demanded. Implicitly, relative prices are taken as given. We will move now to a nonlinear model in which prices are determined given technology and a distributive variable. This type of

model was pioneered by David Ricardo and later formalized by Piero Sraffa⁵ One of its main goals is to allow us to study issues of income distribution between wages and profits.

Let's define v = value of intermediate inputs, π = profits, r = profit rate, w = wage cost, and p = price. We can write:

$$v + \pi + w = p .$$

This equation simply requires that the price is equal to the total cost which is the sum of the three elements of cost, namely intermediate goods, capital and labor. Then assuming that profits are equal to the profit rate times the value of the intermediate inputs we have:

$$v + vr + w = p$$

or:

$$v(1+r) + w = p .$$

Thus, a simple three-good production prices model can be formalized as:

$$\begin{aligned} (a_{11} p_1 + a_{21} p_2 + a_{31} p_3) (1+r) + l_1 w &= p_1 \\ (a_{12} p_1 + a_{22} p_2 + a_{32} p_3) (1+r) + l_2 w &= p_2 \\ (a_{13} p_1 + a_{23} p_2 + a_{33} p_3) (1+r) + l_3 w &= p_3 \end{aligned}$$

where the a 's are, as before, input-output coefficients⁶ and where the l 's are also input-output coefficients indicating the quantity of labor required for the production of one unit of product, p are relative prices, w is the wage per unit of labor (assumed to be uniform for the whole economy), and r is the profit rate. The profit rate is the same for every industry, implying that we are dealing with a long run situation in which capital earns the same profit no matter the industry. Otherwise there would be capital movements from industries with a low rate to industries with a higher rate until that rate equalizes across industries.

The model above has five variables and three equations. Since all prices are relative prices, we need to choose a numeraire, either fixing one variable (say, one price) or introducing a restriction involving some variables. Once we have done this, to close the system of equations we are still left with a degree of freedom regarding w and r . We can thus fix, for example, the unit wage w . A GAMS representation of this model is provided below, where we have chosen a particular set of values for the input-output coefficients, and where we set $p_1 = 1$ and $w = 0$.

⁵ Ricardo (1951) and Sraffa (1972).

⁶ Notice that subscripts of the a 's input-output coefficients are reversed, that is, the input-output A matrix is, in general terms, the transposed of the A matrix corresponding to input-output Leontieff type models. This is so because here we determine prices given technology, while in Leontieff models we determine quantities given technology. To learn more about this, see Passinetti (1977).

```

$title ProdPri
* Production Prices Model

SCALARS
L1 /0.2/
L2 /0.5/
L3 /0.3/;

VARIABLES
p1
p2
p3
w
r
j performance index;

EQUATIONS
eqp1
eqp2
eqp3
jd performance index definition;

jd.. j =E= 0;
eqp1.. (0.3*p1 + 0.1*p2 + 0.4*p3) * (1+r) + L1 * w =E= p1;
eqp2.. (0.2*p1 + 0.4*p2 + 0.1*p3) * (1+r) + L2 * w =E= p2;
eqp3.. (0.2*p1 + 0.5*p2 + 0.2*p3) * (1+r) + L3 * w =E= p3;

w.fx = 0;
p1.fx = 1;

MODEL PP1 /all/;
SOLVE PP1 MAXIMIZING J USING NLP;
DISPLAY p1.l, p2.l, p3.l, w.l, r.l;

```

The solution for r is 0.25. It is interesting to observe what happens as we decrease r . To do so, we now set r equal to different fixed values, that is, we substitute $r.fx = 0.25$ (and later $r.fx=0.20$, etc) for $w.fx = 0$ in the GAMS representation above. We will find that there is an inverse relationship between the unit wage w and the profit rate r , such as the one shown in the table below.

r	w
0.275	0
0.20	0.157
0.15	0.27
0.10	0.389
0.05	0.515
0	0.648

Concerning changes in prices, for this particular example and experiment they will go up as r decreases. But in general prices can go either way (some may go up, others down) depending of the technology, that is the input-output coefficients. However, if we choose w as the numeraire, we will observe that as r increases, all prices increase, indicating that the real wage will decrease no matter the weights used to compute the corresponding wage deflator.

Another interesting experiment would be to pick one price as the numeraire (say $p_1 = 1$, as we did above) and a technology such that the proportions between labor costs and total input costs is the same for each industry, that is, when the input-output coefficients are proportional for all industries.⁷ In this case we will observe that prices will not change as r and w change in a inverse relationship.

3. General Equilibrium Model

In this section we move to a model in which prices and quantities are determined simultaneously. General Equilibrium models of this type were pioneered by Leon Walras and generalized by Nobel prize winners Kenneth Arrow and Gerard Debreu.⁸ One of its main goals is the study of changes in prices and quantities when technology, preferences or endowments change.

Imagine that we have a very simple economy, with only one production sector, two factors of production and a single household. The production sector produces a single good q_s with a Cobb-Douglas constant returns to scale production technology using two inputs: labor and capital. Technical progress (b) can affect total factor productivity. The corresponding labor and capital demand functions (l_d and k_d) are derived combining the production function with the assumption of profit maximizing behavior. Labor and capital supplies (l_s and k_s) are given exogenously. The single household provides labor and capital in exchange for the corresponding wage (w) and profit (r), spending all its income (y) in the demand for the single good (q_d). So far, we have three markets: labor, capital and good markets, and we impose market clearing conditions specifying that supply equals demand. The model equations are listed below:

production function:
$$q_s = b l_d^a k_d^{1-a}$$

⁷ For instance, when matrix $A = \begin{bmatrix} 0.05 & 0.025 & 0.1 \\ 0.1 & 0.05 & 0.2 \\ 0.2 & 0.1 & 0.4 \end{bmatrix}$ and the labor coefficients vector $L = \begin{bmatrix} 1/7 \\ 2/7 \\ 4/7 \end{bmatrix}$.

⁸ Walras (1969), Debreu (1986), Arrow and Hahn (1971).

labor demand, supply and market clearing:

$$l_d = \frac{a q_s p}{w}, \quad l_s = \bar{l}_s, \quad l_s = l_d$$

capital demand, supply and market clearing:

$$k_d = \frac{(1-a) q_s p}{r}, \quad k_s = \bar{k}_s, \quad k_s = k_d$$

household income: $y = w l_d + r k_d$

good demand: $q_d = \frac{y}{p}$

good market clearing: $q_s = q_d$

This simple model has 7 variables and 7 equations. However, one of them is redundant, since “Walras law” establishes that for n-markets we need n-1 equilibrium conditions only. Also, since this model determines relative prices (p , w and r), we need to fix one of them as the numeraire. Thus, by choosing one price as the numeraire (say we fix $p = 1$) and deleting the corresponding good market clearing equation, we are left with a 6-variable 6-equation well defined model. The GAMS representation of the model is shown below. Arbitrary numbers have been chosen for the parameters and for the labor and capital stocks.

\$TITLE SIMPLEGE

SCALARS

a labor share / 0.7 /

b technology parameter / 1.2 /;

POSITIVE VARIABLES

qs good supply

qd good demand

ld labor demand

ls labor supply

kd capital demand

ks capital supply

p price

w wage

r profit

y income;

VARIABLES

j performance index;

EQUATIONS

eqs good supply equation (production function)

eqd good demand equation

```

eld labor demand equation
els labor supply equation
ekd capital demand equation
eks capital supply equation
ey income equation
eml labor market clearing
emk capital market clearing
jd performance index definition;

jd..    j =E= 0;

eqs..   qs =E= b * ld**a * kd**(1-a);
eld..   ld =E= a * qs * p / w;
els..   ls =E= 2;
eml..   ld =E= ls;
ekd..   kd =E= (1-a)* qs * p / r;
eks..   ks =E= 1;
emk..   kd =E= ks;
ey..    y =E= w * ld + r * kd;
eqd..   qd =E= y / p;

*lower bounds to avoid division by zero
p.lo = 0.001; w.lo = 0.001; r.lo = 0.001;

*numeraire
p.fx = 1;

MODEL SIMPLEGE /all/;
SOLVE SIMPLEGE MAXIMIZING J USING NLP;
DISPLAY qs.l, qd.l, ld.l, ls.l, kd.l, ks.l, p.l, w.l, r.l, y.l;

```

The solution values are:

```

qs.L = 1.949 good supply
qd.L = 1.949 good demand
ld.L = 2.000 labor demand
ls.L = 2.000 labor supply
kd.L = 1.000 capital demand
ks.L = 1.000 capital supply
p.L = 1.000 price
w.L = 0.682 wage
r.L = 0.585 profit
y.L = 1.949 income

```

It is important to perform some basic checks on the workings of the model. For instance, since we assumed market clearing, we have to verify that supply equal demand in each market. Also, when increasing the value of the numeraire, all quantity variables should remain the same, while nominal variables (prices and income) should increase proportionally.

Some interesting experiments can be performed with this model. The economy-wide effects of technological progress can be simulated by increasing the value of the b

parameter. Also, we could change the supply of labor or the supply of capital and see how wage and profits are affected. If we do so, we will observe that, in our simple model, that quantities do not change, only the wage and the profit rate do. Quantities would change if we specify elastic labor and capital supply functions, instead of the fixed supplies we assumed.

Also, we imposed the market clearing condition in all three markets. However, it may well be the case that that condition may not be appropriate for some markets because they are in “disequilibrium”. That may happen, for example, because their prices are exogenously fixed. For such cases we should follow an appropriate modeling strategy.⁹

4. Computable General Equilibrium Models

So far we have presented very small models. However, applied economy-wide models tend to be large, thus making the use of computational techniques unavoidable. In this section we will introduce a slightly larger model than the General Equilibrium model presented in section three, to have a flavor of what is like to deal with more than a handful of variables and equations. Models like this are known in the literature as Computable General Equilibrium (CGE) models.¹⁰ We will later go back to a small model to illustrate the application of a linearization technique useful when dealing with relatively large nonlinear models.

4.1 A SAM Based Model

We will move now to a two-sector, two-factor, two-household model to illustrate how to build a CGE model based on a Social Accounting Matrix (SAM). This model was developed by Arne Drud at the World Bank and is discussed in Kendrick (1990).¹¹

Following the research of Nobel prize winner Richard Stone¹², a SAM contains information on the flow of goods and payments between institutions in the economy. Below we present a simple one where the table should be read following the principle that columns pay rows and where each column adds up to the same number as the corresponding row.

⁹ See for example Malinvaud (1977).

¹⁰ For a historical and analytical introduction to this topic, see Dixon and Parmenter (1996). For extended textbook presentations, see Dervis et al. (1982) and Dixon et al. (1992).

¹¹ They implemented the model in Hercules, a system which allows the modeler to develop CGE models by providing basic information in the form of Social Accounting Matrices and by choosing from a menu the functional forms for production functions and demand functions. GAMS provides now a solver (MPSGE) which performs similar functions to those of Hercules (see www.gams.com). These type of systems for model representation are very useful and specially time saving for the experienced modeler. However, here we will present a direct GAMS representation of the Drud-Kendrick which is more suitable to introduce beginners to basic issues in computational model building.

¹² Stone (1961).

	Factors		Households		Sectors	
	Labor	Capital	Rural	Urban	Food	Clothing
Factors						
Labor					75	85
Capital					50	60
Households						
Rural	90	30				
Urban	70	80				
Sectors						
Food			60	65		
Clothing			60	85		

Usually, a SAM can be constructed using a country's official statistics such as the national accounts. Based on the table above, Drud and Kendrick built the following model:

	Quantity	Price, Share or Payment	Price-Quantity
	q	p	pq
Sectors			
Output	$q_s = b_s \prod_f c_{fs}^{a_{fs}}$		$y_s = p_s q_s$
Input	$c_{fs} = \frac{a_{fs} q_s p_s}{p_f}$		$t_{fs} = p_f q_{fs}$
Factors			
Income			$y_f = p_f q_f$
Transfer		$t_{hf} = a_{hf} q_f$	
Household			
Consumption		$t_{sh} = a_{sh} q_h$	$t_{sh} = p_s c_{sh}$
CPI		$p_h = \prod_s p_s^{a_{sh}}$	$y_h = p_h q_h$
Linkage			
Sectors		$y_s = \sum_h t_{sh}$	
Factors		$y_f = \sum_s t_{fs}$	
Households		$y_h = \sum_f t_{hf}$	

The model contains three types of key variables: price (p), quantity (q) and income (y), all of them with a single subscript since they apply to a single institution (subscript f indicates factor, h household and s sector). There are also two additional types of variables: payment (t) and commodity (c), with two subscripts since they represent flows of goods and payment. The subscripts on the payment variables follow the SAM convention: payments are from columns to rows (i.e. t_{fs} indicates payment from s to f). Commodity flows follow a mixed convention (i.e. c_{fs} indicates the amount of factor f used in sector s , while c_{hs} is the flow of purchased goods from sector s to household h).

The output-quantity equations specify production functions with a Cobb-Douglas technology where b is a technology parameter, while the input-quantity equations are the corresponding factor demand equations derived from the corresponding production functions and imposing a zero profit condition. The CPI-price equations are price indexes for the rural and urban households respectively. The a 's are share parameters derived from the SAM.

When expanded, the model has 38 variables and 36 equations. Taking the amount of labor and capital as given (that is, as exogenous variables), choosing one price as the numeraire (say we fix $p_{(urban)} = 1$) and deleting the corresponding market clearing equation (in this case, deleting the linkage equation $y_{(urban)} = \sum_f t_{(urban,f)}$ will do the job), we are left with a model with 36 endogenous variables and 36 equations. The GAMS representation of this model is shown below.

```

$TITLE SAM-Drud-Kendrick
options limrow = 4;

SETS
i general index /labor, capital, rural, urban, food, clothing/
s(i) sectors /food, clothing/
f(i) factors /labor, capital/
h(i) households /rural, urban/;
ALIAS (i,ip);
ALIAS (i,iq);

PARAMETERS
b(s) technical coefficients
a(i,ip) share coefficients;
b('food') = 1.2; b('clothing') = 1;

TABLE sam(i,ip)
      labor capital rural urban food clothing
labor          75  85
capital        50  60
rural    90   30
urban    70   80
food          60  65
clothing      60  85 ;

```

$a(i,ip) = \text{sam}(i,ip) / \text{sum}(iq, \text{sam}(iq,ip));$
 DISPLAY a;

POSITIVE VARIABLES

p(i) price
 q(i) quantity
 y(i) income
 t(i,ip) payment
 c(i,ip) commodity ;

VARIABLES

j performance index;

EQUATIONS

eph(h)
 eqs(s)
 eys(s)
 eyf(f)
 eyh(h)
 etfs(f,s)
 ethf(h,f)
 etsh(s,h)
 eetsh(s,h)
 ecfs(f,s)
 eeys(s)
 eeyf(f)
 eeyh(h)
 jd performance index definition;

* performance index equation
 jd.. $j = E = 0;$

*sectors

eqs(s).. $q(s) = E = b(s) * \text{prod}(f, c(f,s) ** a(f,s));$
 ecfs(f,s).. $c(f,s) = E = a(f,s) * q(s) * p(s) / p(f);$
 eys(s).. $y(s) = E = p(s) * q(s);$
 etfs(f,s).. $t(f,s) = E = p(f) * c(f,s);$

*factors

eyf(f).. $y(f) = E = p(f) * q(f);$
 ethf(h,f).. $t(h,f) = E = a(h,f) * y(f);$

*households

etsh(s,h).. $t(s,h) = E = a(s,h) * y(h);$
 eph(h).. $p(h) = E = \text{prod}(s, p(s) ** a(s,h));$
 eetsh(s,h).. $t(s,h) = E = p(s) * c(s,h);$
 eyh(h).. $y(h) = E = p(h) * q(h);$

*linkage

eeys(s).. $y(s) = E = \text{sum}(h, t(s,h));$
 eeyf(f).. $y(f) = E = \text{sum}(s, t(f,s));$
 eeyh('rural').. $y('rural') = E = \text{sum}(f, t('rural',f));$

*notice that we eliminate one of the linkage equations (Walras law)

*initial values to facilitate solver convergence

```

p.l(i) = 1; q.l(i) = 1; y.l(i) = 1;

*lower bound to avoid division by zero
p.lo(f) = 0.001;

*lower bounds to avoid undefined derivatives in exponential functions
p.lo(s) = 0.001; c.lo(f,s) = 0.001;

*exogenous variables
q.fx('labor') = 2; q.fx('capital') = 1;

*numeraire
p.fx('urban') = 1;

MODEL SAMDK /all/;
option iterlim = 10000;
SOLVE SAMDK MAXIMIZING J USING NLP;

PARAMETER REPORT;
REPORT(i, "price") = p.l(i);
REPORT(i, "quantity") = q.l(i);
REPORT(i, "income") = y.l(i);

DISPLAY REPORT; DISPLAY t.l, c.l;

```

The GAMS representation is similar to the simple General Equilibrium model presented before. Here we make use of sets and subsets as indices, we use the ALIAS command to redefine and index so we can use it to index a matrix, we input the SAM as a table under the PARAMETER section, and we define indexed variables and equations.¹³

As in the previous example, we should check that only nominal variables change (proportionally) when we change the numeraire. And we can perform interesting experiments by changing the amount of labor or capital or the technology parameters. Also, an interesting exercise would be to expand the model to incorporate foreign trade as in Kendrick (1990).

4.2 A Johansen Style Model

CGE models tend to be large and nonlinear. As they grow in size, obtaining convergence (that is, a numerical solution) is likely to become more and more difficult. An alternative is to switch to a model representation pioneered by Leifh Johansen.¹⁴ Johansen style models are solved in a linearized form where all the variables are rates of growth.¹⁵

¹³ Notice that, in order to have a more compact representation, we were able to use a general index “i” for variables, and later work with subsets of variables, but we did not do so for equations. GAMS does not admit the use of subsets as indices of equations.

¹⁴ Johansen (1960).

¹⁵ Notice that, at variance with the nonlinear model, for the linear version the superposition principle will apply: the combined effect of changes in more than one exogenous variable will be equal to the sum of the individual effects

This method consists in transforming all the variables in the model into percentage changes with respect to a base case.¹⁶

For example, given an expression in levels like:

$$X = a Y Z$$

if we first take logs, we obtain:

$$\log X = \log a + \log Y + \log Z$$

and totally differentiating:

$$d(\log X) = d(\log a) + d(\log Y) + d(\log Z)$$

that is:

$$\frac{dX}{X} = \frac{dY}{Y} + \frac{dZ}{Z}$$

or:

$$x = y + z$$

where x , y and z variables are percentage deviations. In a similar fashion, we can transform:

$$X = a Y^b$$

into:

$$x = b y.$$

For an expression like:

$$X = Y + Z$$

we totally differentiate:

$$dX = dY + dZ$$

then divide by the right hand side variable:

$$\frac{dX}{X} = \frac{dY}{X} + \frac{dZ}{X}$$

¹⁶ For CGE Johansen style modeling techniques, see Dixon et al (1992) and Kendrick (1990). For an application of these techniques to a dynamic macroeconomic model, see Mercado and Kendrick (1997).

then multiply and divide each term on the right hand side by the variable on its numerator and rearrange to obtain:

$$\frac{dX}{X} = \frac{dY}{Y} \frac{Y}{X} + \frac{dZ}{Z} \frac{Z}{X}$$

or:

$$x = s_y y + s_z z$$

where s_y, s_z are the shares $s_y = \frac{Y}{Y+Z}$ and $s_z = \frac{Z}{Y+Z}$.

In short, the transformation of a model in levels into one in percentage changes can, in many cases, be achieved by applying some simple rules. Given X, Y and Z as variables in levels, a and b as parameters and x, y and z as variables in percentage deviations, some useful rules are:

$$X = a Y Z \quad \text{becomes} \quad x = y + z$$

$$X = a Y^b \quad \text{becomes} \quad x = b y$$

$$X = Y + Z \quad \text{becomes} \quad x = s_y y + s_z z$$

where s_y, s_z are the shares $s_y = \frac{Y}{Y+Z}$ and $s_z = \frac{Z}{Y+Z}$.

Applying these rules to the simple General Equilibrium model presented in section 3, and now interpreting each variable not as levels but as percentage changes with respect to a base case, we obtain the following GAMS representation:

```

$TITLE JohansenGE
SCALARS
a labor share / 0.7 /
VARIABLES
qs good supply
qd good demand
ld labor demand
ls labor supply
kd capital demand
ks capital supply
p price
w wage
r profit
y income
j performance index;

```

```

EQUATIONS
eqs good supply equation (production funcion)
eqd good demand equation
eld labor demand equation
els labor supply equation
ekd capital demand equation
eks capital supply equation
ey income equation
eml labor market clearing
emk capital market clearing
jd performance index definition;

jd..    j =E= 0;

eqs..   qs =E= ld * a + kd *(1-a);
eld..   ld =E= qs + p - w;
els..   ls =E= 0;
eml..   ld =E= ls;
ekd..   kd =E= qs + p - r;
eks..   ks =E= 0;
emk..   kd =E= ks;
ey..    y =E= (0.7)*(w + ld)+ 0.3 *(r + kd);
eqd..   qd =E= y - p;

*numeraire
p.fx = 0;

MODEL JOHANSENGE /all/;
SOLVE JOHANSENGE MAXIMIZING J USING LP;
DISPLAY qs.l, qd.l, ld.l, ls.l, kd.l, ks.l, p.l, w.l, r.l, y.l;

```

Notice that we eliminated the b parameter from the scalars section, since we do not use it here. Also, notice that we no longer define the model variables as positive variables as we did in the version of the model where variables were in levels, since percentage changes can be positive or negative. Finally, notice that the values of the stock of labor and capital and the numeraire are equal to zero, since they are percentage changes. The 0.7 and 0.3 coefficients that appear in equation “ey” are the corresponding share parameters obtained when applying the third rule. Finally, we solve the model invoking a Linear Programming solver, since the problem is a linear one.

An interesting exercise is to compare the results of the nonlinear model in levels versus the linear model in percentage changes for a given change in an exogenous variable. For example, say that we increase the stock of capital by 20 percent. This means that in the nonlinear model k goes from 1 to 1.2, while in the linear model it goes from zero to 0.2. The results are shown in the table below:

variable	Nonlinear Model		Linearized Model	
	solution k = 1	solution k = 1.2	percentage change	percentage change
q	1.949	2.059	5.6	6
l	2	2	0	0
k	1	1.2	20	20
w	0.682	0.721	5.7	6
r	0.585	0.515	-12	-14
y	1.949	2.059	5.6	6

The differences between the last two columns give us an idea of the approximation error of the linearized solution. We should expect this error to be larger the larger the change in the exogenous variables.

As we said above, solving nonlinear models may become problematic as they grow in size. The problem we just linearized using Johansen's technique is a very small one, and we used it to provide a simple illustration of the methodology. For an application to a larger model you are referred to Kendrick (1990), who develops a Johansen style GAMS representation of a version of the ORANI model developed by the Project Impact in Melbourne, Australia.

References

- Arrow, K. and F. Hahn (1971), *General Competitive Analysis*, Holden-Day.
- Brooke A., D. Kendrick, A. Meeraus y R. Raman (1998), *GAMS: a user's guide*, GAMS Development Corporation (www.gams.com).
- Chiang, A. (1984), *Fundamental Methods of Mathematical Economics*, Third Edition, McGraw-Hill.
- Debreu, G. (1986), *Theory of Value: an Axiomatic Analysis of Economic Equilibrium*, Yale University Press.
- Dervis, K., J. de Melo and S. Robinson (1982), *General Equilibrium Models for Development Policy*, Cambridge University Press.
- Dixon, P. and B. Parmenter (1996) "Computable General Equilibrium Modelling for Policy Analysis and Forecasting" in H. Amman, D. Kendrick and J. Rust, *Handbook of Computational Economics*, Volume 1, Elsevier.
- Dixon, P., B. Parmenter, A. Powell and P. Wilcoxon (1992), *Notes and Problems in Applied General Equilibrium Analysis*, North Holland.
- Johansen, L. (1960), *A Multi-Sectoral Model of Economic Growth*, North Holland Publishing Company.
- Kendrick, D. (1990), *Models for Analyzing Comparative Advantage*, Kluwer Academic Publishers.
- Leontief, W. (1953), *Studies in the Structure of the American Economy*, Oxford University Press.
- Malinvaud, E. (1977), *The Theory of Unemployment Reconsidered*, Basil Blackwell.
- Mercado, P. R. and D. Kendrick (1997), *HTGAMS: Hall and Taylor's Model in GAMS*, Working Paper 9704, Department of Economics, The University of Texas at Austin.
- Passinetti, L. (1977), *Lectures on the Theory of Production*, Columbia University Press.
- Ricardo, D. (1951), *Principles of Political Economy and Taxation*, in *The Works and Correspondence of David Ricardo*, Volume 1, Edited by P. Sraffa, Cambridge University Press for the Royal Economic Society.
- Sraffa, P. (1972), *Production of Commodities by Means of Commodities*, Cambridge University Press.
- Stone, J. R. N, (1961), *Input-Output National Accounts*, OECD.

Walras, L. (1969), *Elements of Pure Economics*, Augustus M. Kelley Publishers.

Zenios, S. (1996), “Modeling Languages in Computational Economics: GAMS” in H. Amman, D. Kendrick and J. Rust, *Handbook of Computational Economics*, Volume 1, Elsevier.